

SIMATIC

S7-200 Примеры


Группа	Пример к теме
4	Запуск звезда-треугольник с ответным сообщением


Краткое описание

Данный пример управляет запуском звезда-треугольник асинхронного двигателя. Пример No. 12 дополнен ответным сообщением действительного состояния переключателей. Если, например, переключатель звезды дефектен, то это будет опознано с ответным сообщением и SIMATIC S7-200 не переключится через время задержки 5 сек в режим треугольника, исключая тем самым возможность поломки.

Двигатель запускается после нажатия кнопки ВКЛ на входе E0.0 в режиме звезды. По прошествии установленного времени двигатель переключается в режим треугольник. Возможное сообщение об ошибке индицируется посредством лампы на выходе A0.3.

Схема включения



Структура программы**Описание программы вкл. листинг**

Двигатель запускается после нажатия кнопки ВКЛ на входе E0.0 в режиме звезды. По прошествии установленного времени 5 сек двигатель переключается в режим треугольник, если отсутствует сообщение об ошибке переключателей. Сообщение об ошибке индицируется лампой на выходе A0.3. Кнопкой квитирования на входе E0.6 можно сбросить сообщение об ошибке, в случае если ошибка устранена. Ответные сообщения переключателей подаются на входы с E0.3 по E0.5.

Если нажата кнопка ВЫКЛ или сработал автомат защиты двигателя, находящиеся на входах E0.1 и E0.2, двигатель отключается. Если кнопки ВЫКЛ и ВКЛ нажаты одновременно, двигатель остается выключенным.

Части программы "Блокировка и деблокировка", "Включение" и "Включение переключателя треугольника" полностью совпадают с примером No.12, поэтому для них не делается пояснений. Части программы "Включение переключателя звезда", "Запуск таймера переключения" и "Включение переключателя сети" дополнены условием, что отсутствует сообщение об ошибке (A0.3), что позволит установить меркер для переключателя звезды (A0.1) и переключателя сети (A0.0) и запустить таймер переключения (T37).

Появилась новая часть программы "Сообщение об ошибке". В принципе обработка сообщения действует таким образом, что сравнивается состояние выходных сигналов со входами, на которые подано действительное состояние переключателей.


Состояние этих выходов сравнивается с состоянием входа E0.3 для переключателя сети, E0.4 для переключателя звезды и E0.5 для переключателя треугольника. При отклонении будет включен таймер T38 с временем задержки 2 сек, соответствующим максимальному времени переключения автоматов.

Если по прошествии данного времени состояния все еще различаются, устанавливается выход ошибки A0.3. Кнопкой квитирования на входе E0.6 он может быть снова сброшен.

Размер программы составляет 70 слов.

KOP (S7-MicroDOS)	AWL (IEC)
Основная программа	
// TITEL = ЗАПУСК ЗВЕЗДА-ТРЕУГОЛЬНИК С ОТВЕТНЫМ СООБЩЕНИЕМ // E0.0 Кнопка вкл. // E0.1 Кнопка выкл. Нормально замкнута // E0.2 Защита двигателя Нормально замкнута // E0.3 // E0.4 RM. Переключатель звезды // E0.5 RM. Переключатель треугольника // E0.6 Кнопка квитирования // A0.0 Переключатель сети // A0.1 Переключатель звезды // A0.2 Переключатель треугольника // A0.3 Лампа ошибки // T37 Таймер переключения 5 сек // T38 Таймер сообщения об ошибке 2 сек	
// ***** БЛОКИРОВКА *****	
1	<pre> 1 --- E0.1 --- / E0.0 ---(M10.0 K1 S) </pre>
LDN E0.1 // Кнопка Выкл нажата U E0.0 // Кнопка Вкл нажата S M10.0 ,1 // Вспомогательный меркер для блокировки	
2	<pre> 2 --- E0.1 --- E0.0 ---(M10.0 K1 R) </pre>
LD E0.1 // Кнопка Выкл не нажата UN E0.0 // Кнопка Вкл не нажата R M10.0 ,1 // Блокировка снята	

// ***** ВКЛЮЧЕНИЕ *****




```

LD      E0.0      // Кнопка Вкл нажата
UN      A0.0      // Нет переключателя сети
LD      A0.0      // Переключатель сети
U       A0.1      // Переключатель звезды
OLD
LD      E0.2      // Защита двигателя ОК
U       E0.1      // Кнопка ВЫКЛ не нажата
UN      A0.2      // Нет переключателя треугольника
ULD
=       M11.0     // Вспомогательный меркер

```

// ***** ВКЛЮЧЕНИЕ ПЕРЕКЛЮЧАТЕЛЯ ЗВЕЗДЫ *****




```

LD      M11.0     // Вспомогательный меркер
UN      M10.0     // Нет блокировки
UN      T37       // Время переключения не закончилось
UN      A0.3      // Нет сообщения об ошибке
=       A0.1      // Переключатель звезды

```

// ***** ЗАПУСК ТАЙМЕРА ПЕРЕКЛЮЧЕНИЯ *****




```

LDN     M10.0     // Нет блокировки
U       M11.0     // Вспомогательный меркер
UN      A0.3      // Нет сообщения об ошибке
TON     T37,50    // Запуск таймера переключения (5 сек)


```

// ***** ВКЛЮЧЕНИЕ ПЕРЕКЛЮЧАТЕЛЯ СЕТИ *****



```
LD      E0.1      // Кнопка ВЫКЛ не нажата
U       E0.2      // Защита двигателя ОК
UN      M10.0     // Нет блокировки
UN      A0.3      // Нет сообщения об ошибке
LD      E0.0      // Кнопка Вкл нажата
U       A0.1      // Переключатель звезды
O       A0.0      // Переключатель сети
ULD
=       A0.0      // Переключатель сети
```

// ***** ВКЛЮЧЕНИЕ ПЕРЕКЛЮЧАТЕЛЯ ТРЕУГОЛЬНИКА *****



```
LD      A0.0      // Переключатель сети
UN      A0.1      // Переключатель звезды
=       A0.2      // Переключатель треугольника
```


// ***** СООБЩЕНИЕ ОБ ОШИБКЕ *****



```

LD      A0.0      // Переключатель сети
UN      E0.3      // RM. Переключатель сети
LDN     A0.0      // Нет переключателя сети
U       E0.3      // RM. Переключатель сети
OLD
LD      A0.1      // Переключатель звезды
UN      E0.4      // Нет RM. Переключателя звезды
OLD
LDN     A0.1      // Нет переключателя звезды
U       E0.4      // RM. Переключатель звезды
OLD
LD      A0.2      // Переключатель треугольника
UN      E0.5      // Нет RM. Переключателя треугольника
OLD
LDN     A0.2      // Нет переключателя треугольника
U       E0.5      // RM. Переключатель треугольника
OLD
TON     T38,20    // Таймер задержки для сообщения об ошибке (2 сек)


```



```

LD      T38
S       A0.3, 1    // Ошибка


```



```

LD      E0.6      // Кнопка квитирования
R       A0.3, 1

```



```

MEND

```

Указания по преобразованию

Для того чтобы преобразовать TOOLITE2 AWL в S7-Micro/DOS AWL

- Установите 'K' перед каждым числом, не являющимся 16-ричной константой (напр. 4 → K4)
- Замените '16#' на 'KH' для всех 16-ричных констант (напр. 16#FF → KHFF)
- Поставьте запятые для смены полей. Используйте клавиши перемещения или клавишу TAB для перехода от поля к полю.
- Для преобразования программы S7-Micro/DOS AWL в KOP-форму нужно начинать каждый сегмент словом 'NETWORK' и номером. Каждый сегмент в этом примере имеет свой номер на диаграмме KOP. Используйте NWENFG в меню редактора для ввода нового сегмента. Команды MEND, RET, RETI, LBL, SBR и INT требуют отдельных сегментов.

Общие указания

Примеры SIMATIC S7-200 предоставляются заказчику бесплатно. Данные примеры не привязаны к конкретной задаче и являются общей информацией о возможностях применения S7-200. Решение заказчика может отличаться от приведённого здесь.

За правильную работу системы заказчик несёт ответственность сам. Мы обращаем Ваше внимание на действующие нормы Вашей страны и предписания по установке соответствующей системы. Ошибки и изменения возможны.